

# About the Formal Verification of Shannon's Theorems

Reynald Affeldt

Research Institute for Secure Systems

National Institute of Advanced Industrial Science and Technology

The proof of the four color theorem by Appel and Haken in 1976 [3] is a good example of the usefulness of computers in the conception of mathematical proofs. At first, some mathematicians disregarded this proof because it relied on an enumeration by a computer program that cannot be checked by hand. Yet, there are more recent examples of proofs, such as the proof of the Kepler conjecture by Hales [7], that are anyway considered uncheckable with 100% certainty by even the most qualified experts. Computer-driven interactive proofs recently emerged as the solution to guarantee the correctness of such large mathematical proofs.

Software for interactive proofs of mathematics take the form of computer languages for the management and the writing of mathematical facts and their proofs. It is important for such software to keep their trusted base as small as possible. This is the design principle behind proof-assistants such as Coq [4]. The kernel of Coq is an implementation of *type theory*, an alternative to set theory for the foundations of mathematics [8]. In this setting, mathematical facts are expressed using types  $T$  and their proofs  $P$  are written as functional programs, with the dedicated notation  $P : T$ . This analogy with programming languages has been known since 1968 as the *Curry-Howard correspondence*. Since there is an algorithm to check that a program matches its type, Coq is able to provide automatic checking of proofs, relieving humans of this burden.

Compared with work cited above, our purpose here is very modest. We just want to illustrate formal verification using an example from coding theory. We will provide a detailed explanation of the formalization of Shannon's source coding theorem (a.k.a. the noiseless coding theorem) using SSREFLECT [5], an extension to Coq that provides a library that was used in particular to verify the Feit-Thompson theorem [6]. The present document is actually an extended version of the first half of [1]. For details about the formalization of the direct part of the noisy channel coding theorem see [1], for the formalization of the converse of the noisy channel coding theorem see [2], for the matching pencil-and-paper presentation see [9].

We display the formalization as it is processed by Coq. (Between `/*` and `*/` are comments; `...` are for elided parts.) Margin comments are tags referring to the proof script. The lecture material is available at [2]. Interactive proof will be demonstrated during the lecture. Various Coq tutorials are available online [4].

tag in script

# 1 Finite Probabilities

## 1.1 Distributions

A probability space will be the powerset of some finite type  $A$  (type `finType` in `SSREFLECT`). An *event* is a subset of the probability space and can therefore be encoded as a boolean predicate, of type `pred A` in `SSREFLECT`. An elementary event is an event that cannot be decomposed further, i.e., a predicate `pred1 a ≈ fun x ⇒ x = a` (for some  $a \in A$ ). A *distribution* associates a positive real to each elementary event s.t. the sum for all elementary events is 1. The probability of an event is the sum of the probabilities of its elementary events.

**Definition 1** (Distribution). Given a finite type  $A$ , a distribution is defined as a positive real-valued (probability mass) function `pmf`, equipped with a proof `pmf1` that its outputs sum to 1:

1

```
Record dist := mkDist {
  pmf  :> A → R+ ;
  pmf1 : Σ_(a in A) pmf a = 1 }.
```

A distribution is thus a `Record` but thanks to the coercion `>`, we can write “ $P\ a$ ” as a function application to represent the probability associated with  $a$ . The construct `Σ_(a in A) F a` is provided by `SSREFLECT`’s bigops theory, that is further illustrated by the next two examples.

**Example 1** (Product Distribution). Let us assume that we are given two distributions  $P_1 : \text{dist } A$  and  $P_2 : \text{dist } B$ . The *product distribution* over  $A \times B$  noted  $P_1 \times P_2$  is defined as follows:

2

```
Definition prod_dist : dist [finType of A * B].
apply mkDist with (fun ab ⇒ P1 ab.1 * P2 ab.2).
- /* RleOf proof */ ...
- /* pmf1 proof */ /* goal: Σ_(a|a ∈ [finType of A * B]) P1 a.1 * P2 a.2 = 1 */
rewrite -(pair_big xpredT xpredT (fun a b ⇒ P1 a * P2 b)) /=.
/* Σ_(a|a ∈ A) Σ_(b|b ∈ B) P1 a * P2 b = 1 */
rewrite -(pmf1 P1).
/* Σ_(a|a ∈ A) Σ_(b|b ∈ B) P1 a * P2 b = Σ_(a|a ∈ A) P1 a */
apply eq_bigr ⇒ a ..
/* Σ_(b|b ∈ B) P1 a * P2 b = P1 a */
rewrite -big_distr /=.
/* P1 a * (Σ_(b|b ∈ B) P2 b) = P1 a */
rewrite pmf1.
/* P1 a * 1 = P1 a */
...
Defined.
```

**Example 2** (Tuple Distribution). Let us assume that we are given a distribution  $P : \text{dist } A$ . The *tuple distribution* over  $A^n$  noted  $P^n$  is defined as follows:

3

```
Definition tuple_dist : dist [finType of n.-tuple A].
apply mkDist with (fun x ⇒ Π_(i < n) P x_i).
- /* RleOf proof */ ...
- /* pmf1 proof */ /* goal: Σ_(t|t ∈ [finType of n.-tuple A]) Π_(i<n) P t_i = 1 */
pose P' := fun (a : 'I_n) b ⇒ P b.
suff : Σ_(f : {ffun 'I_n → A }) Π_(i < n) P' i (f i) = 1.
```

```

...
/*  $\Sigma_{(f : \{ffun 'I_n \rightarrow A\}) \Pi_{(i < n)} P' i (f i) = 1$  */
rewrite -bigA_distr_bigA /= /P'.
/*  $\Pi_{(i < n)} \Sigma_{(a \text{ in } A)} P a = 1$  */
rewrite [X in _ = X](_ : 1 =  $\Pi_{(i < n)} 1$ ); last by ...
/*  $\Pi_{(i < n)} \Sigma_{(j \text{ in } A)} P j = \Pi_{(i < n)} 1$  */
apply eq_bigr  $\Rightarrow$  i _ .
/*  $\Sigma_{(j \text{ in } A)} P j = 1$  */
...

```

Defined.

$n$ -tuple  $A$  is SSREFLECT notation for  $A^n$ . Given a tuple  $t \in n$ -tuple  $A$ ,  $t_i$  is notation for the  $i$ th component of  $t$ , with  $i \in 'I_n$  is a natural strictly smaller than  $n$  (an *ordinal* in SSREFLECT's parlance).

4

**Definition 2** (Probability of an Event). Given a distribution  $P : \text{dist } A$  and an event  $E : \text{pred } A$ , the probability of  $E$  (w.r.t. distribution  $P$ ) is defined as follows:

**Definition**  $\text{Pr } (E : \text{pred } A) := \Sigma_{(a \text{ in } A \mid E a)} P a$ .

5

**Lemma 1.** Some properties of the probability of events for the sake of completeness (they will be used later). ( $[\text{predC } E]$  is the complement of predicate  $E$ .)

**Lemma**  $\text{Pr\_incl } (E E' : \text{pred } A) : (\forall i, E i \rightarrow E' i) \rightarrow \text{Pr } E \leq \text{Pr } E'$ .

**Lemma**  $\text{Pr\_cplt } E : \text{Pr } E + \text{Pr } [\text{predC } E] = 1$ .

## 1.2 Random Variables

The restriction to finite probabilities makes it possible to simplify the definition of random variables.

6

**Definition 3** (Random Variable). A *random variable* (RV) is a distribution over some finite type  $A$  coupled with a real-valued function defined over  $A$ :

**Record**  $\text{rvar } A := \{\text{rv\_dist} : \text{dist } A ; \text{rv\_fun} :> A \rightarrow \mathbb{R}\}$ .

7

The (implicit) probability space is the powerset of  $A$ . We note  $p_X$  the distribution of the RV  $X$ . We note  $\text{Pr}[X = r]$  the following definition:

**Definition**  $\text{pr } (X : \text{rvar } A) r := \text{Pr } p_X [\text{pred } x \mid X x = r]$ .

8

( $[\text{pred } x \mid .. x ..]$  is a SSREFLECT notation for boolean predicates.) We note  $X - m$  the RV that subtracts  $m$  to the output of  $X$ ,  $k \times X$  the RV that multiplies by  $k$  the output of  $X$ ,  $X / n$  the RV that divides by  $n$  the output of  $X$ ,  $X^2$  the RV that squares the output of  $X$ ,  $-\log P$  the RV  $\text{fun } x \Rightarrow -\log(P x)$  (with underlying distribution  $P$ ).

9

**Definition 4** (Expected Value). The *expected value* of a RV  $X$  (over  $A$ ) is noted  $\mathcal{E} X$  and is defined as follows:

**Definition**  $\text{Ex\_alt} := \Sigma_{(a \text{ in } A)} X a * p_X a$ .

10

This is equivalent to the following definition:

**Definition** `img (A : finType) (f : A → R) := undup (map f (enum A)).`  
**Definition** `Ex := Σ_(r ← img X) r * Pr[ X = r ].`

The expected value is to be understood as the average outcome; this will become clearer when we state the weak law of large number (Lemma 5).

The *variance* is a measure of the deviation of the outcome of an experiment from its expected value.

11 **Definition 5.** Given a RV  $X$ , the *variance* of  $X$  is noted  $\mathcal{V} X$  and is defined as follows:

**Definition** `Var := let μ := Ex in Ex ((X - μ) ^ 2).`

12 **Lemma 2.** The variance can also be computed as follows:

**Lemma** `V_alt : Var = Ex (X ^ 2) - (Ex X) ^ 2.`

The Chebyshev Inequality bounds the probability of deviations from the expected value.

13 **Lemma 3 (Chebyshev Inequality).** Given a RV  $X$  over  $A$ , the following holds:

**Lemma** `chebyshev_inequality ε : 0 < ε → Pr p_X [pred a | Rabs (X a - Ex X) ≥ ε] ≤ V X / ε ^ 2.`

*Proof.* Script excerpt:

```
...
/* ε ^ 2 * Pr p_X [pred a | Rabs (X a - Ex X) ≥ ε] ≤ V X */
rewrite /V {2}/Ex_alt.
/* lhs ≤ Σ_(a | a ∈ A) ((X - Ex X) ^ 2) a * p_X ((X - Ex X) ^ 2) a */
rewrite ( _ : p_X ((X - Ex X) ^ 2) = p_X ) //.
/* lhs ≤ Σ_(a | a ∈ A) ((X - Ex X) ^ 2) a * p_X a */
apply Rle_trans with (Σ_(a in A | Rabs (X a - Ex X) ≥ ε) ((X - Ex X) ^ 2) a
* p_X a);
  last by ...
/* ε ^ 2 * Pr p_X [pred a | Rabs (X a - Ex X) ≥ ε] ≤
  Σ_(a | (a ∈ A) ∧ (Rabs (X a - Ex X) ≥ ε)) ((X - Ex X) ^ 2) a * p_X a */
rewrite /Pr big_distr [- ^ 2]lock /= -!lock.
/* Σ_(a | Rabs (X a - Ex X) ≥ ε) ε ^ 2 * p_X a ≤
  Σ_(a | Rabs (X a - Ex X) ≥ ε) ((X - Ex X) ^ 2) a * p_X a */
...

```

□

### 1.3 Sum of Random Variables

14 **Definition 6 (Joint Distribution).** The fact that  $P$  is a *joint distribution* of  $P_1$  (distribution over  $A$ ) and  $P_2$  (distribution over  $A^n$ ) is captured by the following predicate:

$(\forall x, P_1 x = \Sigma_{(t \text{ in } \{ :n+1.-tuple A \} \mid \text{thead } t = x) P t) \wedge$   
 $(\forall x, P_2 x = \Sigma_{(t \text{ in } \{ :n+1.-tuple A \} \mid \text{behead } t = x) P t).$

(thead takes the head of a tuple, behead its tail.) So, when a distribution  $P$  is a joint distribution of two distributions  $P_1$  and  $P_2$ ,  $P_1$  and  $P_2$  are two marginal distributions of  $P$ . Still, a joint distribution is not necessarily a product distribution.

15

**Definition 7** (Independent Random Variables). The fact that a RV  $X$  (over  $A$ ) and a RV  $Y$  (over  $A^n$ ) are independent w.r.t. the joint distribution  $P$  (over  $A^{n+1}$ ) is noted  $X \perp_P Y$  and holds when:

$$\forall x y, \Pr P [\text{pred } xy \mid (X (\text{thead } xy) = x) \wedge (Y [\text{tuple of behead } xy] = y)] = \Pr[X = x] * \Pr[Y = y].$$

16

**Definition 8** (Sum of Two Random Variables). The fact that  $X$  (RV over  $A^{n+2}$ ) is the sum of  $X_1$  (RV over  $A$ ) and  $X_2$  (RV over  $A^{n+1}$ ) is noted  $X = X_1 + X_2$  and is formalized as follows:

$$\text{Definition sum} := \text{joint } p_{X_1} p_{X_2} p_X \wedge X =_i [\text{ffun } x \Rightarrow X_1 (\text{thead } x) + X_2 [\text{tuple of behead } x]].$$

17

**Lemma 4** (Expected Value and Variance of Sum of RVs). The expected value is always linear but the variance only when the RVs are independent:

$$\text{Lemma E\_linear\_2} : X = X_1 + X_2 \rightarrow \mathcal{E} X = \mathcal{E} X_1 + \mathcal{E} X_2.$$

18

$$\text{Lemma V\_linear\_2} : X = X_1 + X_2 \rightarrow X_1 \perp_{p_X} X_2 \rightarrow \mathcal{V} X = \mathcal{V} X_1 + \mathcal{V} X_2.$$

19

**Definition 9** (Sum of Independent Random Variables). We are given a RV  $X$  over  $A^n$  and a  $n$ -tuple  $Xs$  of random variables (each over  $A$ ). The fact that  $X$  is the sum of the RVs in  $Xs$  considered as independent is noted  $X =_i \Sigma Xs$  and is captured by the following predicate:

$$\begin{aligned} \text{Inductive isum\_n} : \forall n, \text{rvar } [\text{finType of } n\text{-tuple } A] \rightarrow n\text{-tuple } (\text{rvar } A) \rightarrow \text{Prop} := \\ | \text{isum\_n\_1} : \forall X, \text{cast\_rv } X =_i \Sigma X \\ | \text{isum\_n\_cons} : \forall n (Ys : n+1\text{-tuple } \_) Y X Z, \\ Y =_i \Sigma Ys \rightarrow Z = X + Y \rightarrow X \perp_{p_Z} Y \rightarrow Z =_i \Sigma [\text{tuple of } X :: Ys] \end{aligned}$$

The weak law of large numbers is the first fundamental theorem of probability. Intuitively, it says that the average of the results obtained by repeating an experiment a large number of times is close to the expected value (thus justifying the interpretation of the latter as an average outcome).

20

**Lemma 5** (Weak Law of Large Numbers).  $P$  is a distribution (over  $A$ ).  $Xs$  are  $n+1$  identically distributed RVs (each over  $A$ , they all have distribution  $P$ ). We note  $\mu$  their common expected value and  $\sigma^2$  their common variance. Let  $X$  be the sum of the  $Xs$  considered independent, i.e.,  $X =_i \Sigma Xs$ . We have:

$$\begin{aligned} \text{Lemma wlln } \varepsilon : 0 < \varepsilon \rightarrow \\ \Pr p_X [\text{pred } t \mid \text{Rabs } ((X / (n+1)) t - \mu) \geq \varepsilon] \leq \sigma^2 / ((n+1) * \varepsilon^2). \end{aligned}$$

The weak law of large numbers says that the average of the outcome of  $X$  gets closer to  $\mu$  with a large  $n$ .

*Proof.* Script excerpt:

```

have HV :  $\mathcal{V} (X / n+1) = \sigma^2 / n+1$ .
...
...
/* Pr  $p_X$  [pred t | Rabs ((1 * / n+1  $\times$  X) t -  $\mu$ )  $\geq \epsilon$ ]  $\leq$ 
 $\mathcal{V} (X / n+1) * / \epsilon^2$  */
have HE :  $\mathcal{E} (X / n+1) = \mu$ .
...
rewrite -{ }HE.
/* Pr  $p_X$  [pred t | Rabs ((1 * / n+1  $\times$  X) t -  $\mathcal{E} (X / n+1)$ )  $\geq \epsilon$ ]  $\leq$ 
 $\mathcal{V} (X / n+1) * / \epsilon^2$  */
have cheby : Pr  $p_X(X / n+1)$ 
[pred t | Rabs (X t / n+1 -  $\mathcal{E} (X / n+1)$ )  $\geq \epsilon$ ]  $\leq \mathcal{V} (X / n+1) / \epsilon^2$ .
...

```

□

## 2 Information Theory

We now come to the definition of *typical sequence*. Let us assume that we are given a source that emits symbols. Intuitively, a typical sequence is a  $n$ -tuple of symbols that is expected to be observed when  $n$  is “large”. For example, a tuple emitted by a binary source that emits 0 with probability  $2/3$  is typical when it contains about two thirds of 0s.

21

The definition of typical sequences relies on the definition of *entropy*.

**Definition 10** (Entropy). The entropy of a distribution  $P$  (over  $\mathbf{A}$ ) is noted  $\mathcal{H}P$  and is defined as  $-\sum_{\mathbf{a} \in \mathbf{A}} P \mathbf{a} * \log(P \mathbf{a})$ .

The entropy is a measure of uncertainty. For instance, a source that always emits the same symbol has entropy 0. A source that emits symbols drawn uniformly from an alphabet of cardinal  $n$  is maximal:  $\log n$ .

22

**Definition 11** (Typical Sequence). Let  $P$  be a distribution over  $\mathbf{A}$ . A  $n$ -tuple  $\mathbf{x}$  is a  $P, \epsilon$ -typical sequence when it satisfies the following predicate:

**Definition** `typ_seq`  $\mathbf{x} := \exp(-n * (\mathcal{H}P + \epsilon)) \leq P^n \mathbf{x} \leq \exp(-n * (\mathcal{H}P - \epsilon))$ .

23

**Definition 12** (Typical Set). Given a natural  $n$ , a distribution  $P$ , and an  $\epsilon$ , we note  $\mathcal{TS} P n \epsilon$  the set of typical sequences, i.e., the set `[set x | typ_seq x]`.

(`[set x | ...]` is SSREFLECT notation for finite sets.)

To establish that typical sequences do have the properties we expect intuitively, we first need to prove the property of *asymptotic equipartition property* (AEP). This is a property about the outcome of several RVs that are independent and identically distributed. Informally, the AEP states that, in terms of probability,  $-(1/n+1) * \log(P^{n+1} \mathbf{x})$  is “close” to the entropy  $\mathcal{H}P$ .

**Lemma 6** (Asymptotic Equipartition Property). We are given a source that emits symbols with a distribution  $P$  (over  $\mathbf{A}$ ) and some  $\epsilon$  ( $0 < \epsilon$ ). We first define the following quantities that will be used as bounds hereafter:

**Definition** `aep_σ2` :=  $\sum_{(a \text{ in } A)} P \ a \ * \ (\log (P \ a))^2 - (\mathcal{H} \ P)^2$ .  
**Definition** `aep_bound`  $\varepsilon$  := `aep_σ2 P / ε3`.

24 The AEP states the following technical property about the probability of emitting a  $n+1$ -tuple:

**Lemma** `aep` : `aep_bound P ε ≤ n+1 →`  
`Pr (Pn+1) [pred t | (0 < Pn+1 t) ∧`  
`(Rabs ((-log (Pn+1) / n+1) t - H P) ≥ ε) ] ≤ ε.`

*Proof.* We change the shape of the rhs of the goal to match the weak law of large numbers:

```
apply Rle_trans with (aep_σ2 P / (n+1 * ε ^ 2)); last first.
...
/* Pr Pn+1 [pred t | 0 < Pn+1 t &
   Rabs ((-log (Pn+1) / n+1) t - H P) ≥ ε] ≤
   aep_σ2 P / (n+1 * ε ^ 2) */
```

25 We prepare several identically distributed RVs  $X_s$  with distribution  $P$ :

**Definition** `map_mlog A n (P : dist A) : n.-tuple (rvar A) :=`  
`[tuple of map (fun X ⇒ -log X) (nseq n P)].`

26 and the RV defined as follows:

**Definition** `sum_map_mlog A (P : dist A) n : rvar [finType of n.-tuple A].`  
`apply (Build_rvar (Pn)).`  
`exact: (fun x ⇒  $\sum_{(i < n)}$  -log (P xi)).`  
**Defined.**

We can show that the latter is the sum of the former (considered as independent), i.e.  
`sum_map_mlog P n+1 =i  $\sum$  map_mlog n+1 P.`

```
move: (mlog_tuple_isum_map_mlog P n) ⇒ Hisum.
/* Hisum : sum_map_mlog P n+1 =i  $\sum$  map_mlog n+1 P */
```

The conclusion comes from the application of the weak law of large numbers:

```
move: (wlln (@E_map_mlog _ P n) (@V_map_mlog _ P n) Hisum Hε) ⇒ law_large.
/* law_large : Pr p_ (sum_map_mlog P n+1)
   [pred t | Rabs ((sum_map_mlog P n+1 / n+1) t - H P) ≥ ε] ≤
   aep_σ2 P / (n+1 * ε ^ 2) */
eapply Rle_trans; last by apply law_large.
/* Pr Pn+1 [pred t | 0 < Pn+1 t &
   Rabs ((-log (Pn+1) / n+1) t - H P) ≥ ε] ≤
   Pr p_ (sum_map_mlog P n+1)
   [pred t | Rabs ((sum_map_mlog P n+1 / n+1) t - H P) ≥ ε] */
apply Pr_incl ⇒ i /=.
...

```

□

27 **Lemma 7** (Typical Set Non-empty). When  $n$  is large enough,  $\mathcal{TS}$  is not empty:

**Definition** `TS_0 (H : aep_bound P ε ≤ n+1) : [finType of n+1.-tuple A]. ...`  
**Lemma** `TS_0_is_typ_seq (k_k0 : aep_bound P ε ≤ n+1) : TS_0 k_k0 ∈  $\mathcal{TS}$  P n+1 ε.`

28

**Lemma 8** (Typical Set Upper-bound). The cardinal of a  $P, n, \varepsilon$ -typical set is upper-bounded as follows:

**Lemma** `TS_sup` :  $| \mathcal{TS} P n \varepsilon | \leq \exp (n * (\mathcal{H} P + \varepsilon))$ .

29

**Lemma 9** (Probability of Typicality). When  $n$  gets large, emitted sequences are more likely to be typical:

**Lemma** `Pr_TS_1` : `aep_bound`  $P \varepsilon \leq n+1 \rightarrow \Pr (P^{n+1} [\text{pred } t \text{ in } \mathcal{TS} P n+1 \varepsilon]) \geq 1 - \varepsilon$ .

### 3 Source Coding Theorem

The source coding theorem (a.k.a. the noiseless coding theorem) is a theorem for data compression. The basic idea is to replace frequent words with alphabet sequences and other words with a special symbol. Let us illustrate this with an example. The combination of two Roman alphabet letters consists of 676 ( $= 26^2$ ) words. Since  $2^9 < 676 < 2^{10}$ , 10 bits are required to represent all the words. However, by focusing on often-used English words (“as”, “in”, “of”, etc.), we can encode them with less than 9 bits. Since this method does not encode rarely-used words (such as “pz”) decoding errors can happen. Given an information source that emits all symbols with the same distribution  $P$ , the source coding theorem gives a theoretical lower-bound (namely, the entropy  $\mathcal{H}P$ ) for compression rates for compression with negligible error-rate.

30

**Definition 13** (Source Code). A  $k, n$ -source code from alphabet  $A$  is a pair of functions:

**Definition** `encT` :=  $k$ -tuple  $A \rightarrow n$ -tuple `bool`.

**Definition** `decT` :=  $n$ -tuple `bool`  $\rightarrow k$ -tuple  $A$ .

**Record** `scode` := `mkScode` { `enc` : `encT` ; `dec` : `decT` }.

The main characteristic of a source code is its rate:

**Definition** `SrcRate` (`sc` : `scode`) :=  $n / k$ .

The rate is the number of bits needed per symbol.

31

**Definition 14** (Error Rate). Let us assume a source of symbols with probability  $P$  (over  $A$ ). The error rate of a  $k, n$ -source code `sc` is noted  $\bar{e}_{src}(P, sc)$  and is defined as:

$\Pr (P^k) [\text{pred } x \mid \text{dec } sc (\text{enc } sc x) \neq x]$ .

**Theorem 1** (Source Coding—Direct Part). Given a source of symbols from the alphabet  $A$  with distribution  $P$ , there exist source codes of rate  $r \in \mathbb{Q}^+$  larger than the entropy  $\mathcal{H}P$  such that the error rate can be made arbitrary small:

32

**Theorem** `source_coding_direct` :  $\forall \lambda, 0 < \lambda < 1 \rightarrow$   
 $\forall r : \mathbb{Q}^+, \mathcal{H} P < r \rightarrow$   
 $\exists k n (sc : \text{scode } A k n), r = \text{SrcRate } sc \wedge \bar{e}_{src}(P, sc) \leq \lambda$ .

33

*Proof.* The first step is to instantiate  $k$  and  $n$ . For that purpose, we introduce the following intermediate constants:



```

Let r := num / den+1.
Definition ε := Rmin (r - ℋ P) λ.
Definition δ := Rmax (aep_bound P (ε / 2)) (2 / ε).

```

34 We instantiate  $k$  such that  $\delta \leq k$ ;  $n$  follows by the definition of the rate of the source code:  $w$

```

Lemma SrcDirectBound d D : { k | D ≤ (k+1 * d+1) }. ...

```

```

Let k' := sval (SrcDirectBound den δ).
Definition k := k'+1 * den+1.
Definition n := k'+1 * num.

```

35 We now construct a generic  $k,n$ -source code encoder. Let  $S$  be a subset of  $A^{k+1}$  such that its cardinal is strictly less than  $2^n$ . The function that associates to the  $i$ th element  $x$  of  $S$  the  $n$ -bit encoding of  $i+1$  in binary and  $0\dots 0$  ( $n$  times) otherwise is defined as follows:

```

Definition f : encT A k+1 n := fun x =>
  if x ∈ S then
    let i := index x (enum S) in Tuple (size_nat2bin_b i+1 n)
  else
    [tuple of nseq n false].

```

36 We now construct a corresponding  $k,n$ -source code decoder. Given a  $n$ -tuple  $x$ , if  $x$  is not 0 and smaller than  $|S|$ , it returns the “ $x-1$ ”th element of  $S$ , and some default element otherwise:

```

Variable def : k+1.-tuple A. Hypothesis Hdef : def ∈ S.

```

```

Definition φ : decT A k+1 n := fun x =>
  let i := tuple2N x in
  if i is 0 then def else
    if i-1 < | S | then nth def (enum S) i-1 else def.

```

37 By construction:

```

Lemma φ_f i : φ (f i) = i ↔ i ∈ S.

```

38 We now instantiate the generic source code above so that it encodes properly the typical sequences (proof script excerpt):

```

set S := ℳ P k (ε / 2).
set def := TS_0 halfepsilon0 halfepsilon1 k_k0.
set F := f n S.
set PHI := @φ _ n _ S def.
exists {| enc := F; dec := PHI |}.

```

The rate of the instantiated source code is the good one by construction. The conclusion of the proof follows from the properties of typical sequences:

```

/* goal(1):  $\bar{e}_{src}(P, \{| enc := F; dec := PHI | \}) \leq \lambda$  */

```

The lhs of goal(1) can be rewritten so as to transform the goal into:

```

/* goal(2):  $1 - \Pr P^k [\text{pred } t \text{ in } \mathcal{TS} P k (\varepsilon / 2)] \leq \lambda$  */

```

The latter is easily provable using the lemma `Pr_TS_1` (Lemma 9) because it can be rewritten as follows:

```
/* Pr P^k [pred t in TS P k (ε / 2)] ≥ 1 - ε / 2 */
```

We therefore only need to prove the rewriting from goal(1) to goal(2), i.e.:

```
/* Σ_(t|PHI (F t) ≠ t) Π_(i<k) P t_i = 1 - Σ_(t|t ∈ TS P k (ε / 2)) Π_(i<k) P t_i */
```

By appealing to `Pr_cpl1t` (probability of the complement event, Lemma 1), the rhs can be rewritten such that:

```
/* Σ_(t|PHI (F t) ≠ t) Π_(i<k) P t_i = Σ_(t|(t ∈ k.-tuple A) ∧ (t ∉ S)) P^k t */
```

which amounts to prove:

```
/* (PHI (F t) ≠ t) = (t ∉ S) */
```

This is essentially the lemma `phi_f` but we need to prove two things to use it. The first one is `def ∈ S`, that comes directly from `TS_0_is_typ_seq` (see Lemma 7). The second one is `|S| < 2^k`. This can be reduced to `1 + |S| ≤ exp(k * (HP + ε))` by using the definition of the source rate and the definition of  $\varepsilon$ . This can be further reduced to `1 + |S| ≤ exp(1 + k * (HP + ε/2))` which can be proven using `TS_sup` (Lemma 8).  $\square$

**Theorem 2** (Source Coding—Converse Part). The converse of the Shannon’s source coding theorem shows that any source code whose rate is smaller than the entropy of a source with distribution  $P$  over  $A$  has non-negligible error-rate:

39

```
Theorem source_coding_converse : ∀ λ, 0 < λ < 1 →
  ∀ r : Q+, 0 < r < H P →
  ∀ n k (sc : scode A k+1 n), r = SrcRate sc →
    SrcConverseBound P (num r) (den r) n λ ≤ k+1 →
      e_src(sc, P) ≥ λ.
```

where the bound `SrcConverseBound` gives a precise meaning to the claim that would otherwise be informally summarized as “for  $k$  big enough”:

40

```
Definition ε := Rmin ((1 - λ) / 2) ((H P - r) / 2).
```

```
Definition δ := Rmin ((H P - r) / 2) (ε / 2).
```

```
Definition SrcConverseBound := Rmax (Rmax
  (aep_bound P δ) (- ((log δ) / (H P - r - δ)))) (n / r).
```

See [2] for the proof and also for the channel coding theorems.

## References

- [1] Reynald Affeldt, Manabu Hagiwara. Formalization of Shannon’s Theorems in `SSReflect-Coq`. In 3rd Conference on Interactive Theorem Proving (ITP 2012), volume 7406 of Lecture Notes in Computer Science, pages 233-249. Springer, August 2012
- [2] Reynald Affeldt, Manabu Hagiwara, Jonas Sénizergues. Formalization of Shannon’s Theorems. <http://staff.aist.go.jp/reynald.affeldt/shannon>. Coq documentation. AIST, 2012
- [3] Kenneth Appel, Wolfgang Haken. Every map is four colourable. *Bulletin of the American Mathematical Society* 82:711–712 (1976)

- [4] The Coq Proof Assistant. <http://coq.inria.fr> INRIA, 1985–2012
- [5] Georges Gonthier, Assia Mahboubi, Enrico Tassi. A Small Scale Reflection Extension for the Coq System. Technical Report 6455. Version 11. INRIA, 2012
- [6] Georges Gonthier. RE: [Coqfinitgroup-commits] r4105 - trunk. Electronic mail. Available at: <http://www.msr-inria.inria.fr/events-news/feit-thompson-proved-in-coq>.
- [7] Thomas Hales. A proof of the Kepler conjecture. *Annals of Mathematics* 162(3):1065–1185 (2005)
- [8] Jean van Heijenoort. *From Frege to Gödel: A Source Book in Mathematical Logic, 1879-1931*. Harvard University Press, 1967
- [9] Manabu Hagiwara. *Coding Theory: Mathematics for Digital Communication*. In Japanese. <http://www.nippyo.co.jp/book/5977.html>. Nippon Hyoron Sha, 2012